# $Id: PKGBUILD 79363 2012-11-03 16:29:50Z lfleischer $
# Maintainer: Lukas Fleischer <archlinux at cryptocrack dot de>
# Contributor: Allan McRae <mcrae_allan@hotmail.com>
# Contributor: fancris3 <fancris3 at aol.com>
# Contributor: Daniel J Griffiths <ghost1227@archlinux.us>

pkgname=gecko-mediaplayer
pkgver=1.0.7
pkgrel=1
pkgdesc='Browser plugin that uses gnome-mplayer to play media in a web browser.'
arch=('i686' 'x86_64')
url='http://code.google.com/p/gecko-mediaplayer/'
license=('GPL')
depends=("gnome-mplayer>=${pkgver}" 'dbus-glib' 'gmtk')
makedepends=('xulrunner' 'pkgconfig')
replaces=('mplayer-plugin')
source=("http://gecko-mediaplayer.googlecode.com/files/${pkgname}-${pkgver}.tar.gz")
md5sums=('d35e3307126f874afba93bd896d605dc')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  ./configure --prefix=/usr --sysconfdir=/etc --without-gconf --enable-new-libxul
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
